
This document lists all the essential changes needed to correct problems in the iwd-TTCN3-B2022-09_D23wk49 related to the title of this CR.
Contact:	Parikshit Bhise
	Parikshit.bhise@rohde-schwarz.com

2 [bookmark: _Toc140667222][bookmark: _Toc158730240]Corrections required
2.1 Correctiion to fl_NBIOT_NAS_CellInfo_Init
	Function name
	fl_NBIOT_NAS_CellInfo_Init

	Reason for change
	According to 8.3.3.1.2 SIB 5 is not available to a number of cells and in the function have each non-compatible cells’s sys info combination changed from c3 to c1. However, this doesn’t take into account the new sys info combination c11 which is the NTN counterpart for c3 and thus also needs a case to turn it into c1’s counterpart c8. This effects the TC 22.4.14 where c3 is used as a combination and is changed to c11 in the init function.


	Summary of change
	Changed the if statement for if sib 5 is present in the sys info combination by adding c11 to the if statement. Additionally added a check to convert the sys info combination to its NTN counterpart.

	TTCN module
	NBIOT_CellInfo.ttcn

	MCC160 Comment
	



Before:

>>> Skipped Code <<<  
 // Initialise NAS parameters
    v_NAS_Param := fl_NBIOT_InitNAS_forNAS_Cells( p_CellId);
    
    // SIB5 is not valid
    if (p_SysinfoCombination == c3) { // @sic R5s170755 sic@
      if (p_CellsOnPLMN == LTE_AllCellsOnSamePLMN and (p_CellId == nbiot_Cell54 or
                                                       p_CellId == nbiot_Cell55 or
                                                       p_CellId == nbiot_Cell56 or  // @sic R5s170511 sic@
                                                       p_CellId == nbiot_Cell57 or
                                                       p_CellId == nbiot_Cell58 or
                                                       p_CellId == nbiot_Cell59 or
                                                       p_CellId == nbiot_Cell60 or
                                                       p_CellId == nbiot_Cell62 or
                                                       p_CellId == nbiot_Cell63 ))
        {v_SysinfoCombination := c1}
      if (p_CellsOnPLMN == LTE_CellsOnDifferentPLMN and (p_CellId == nbiot_Cell50 or
                                                         p_CellId == nbiot_Cell51 or
                                                         p_CellId == nbiot_Cell52 or  // @sic R5s170511 sic@
                                                         p_CellId == nbiot_Cell53 or
                                                         p_CellId == nbiot_Cell55 or
                                                         p_CellId == nbiot_Cell56 or
                                                         p_CellId == nbiot_Cell58 or
                                                         p_CellId == nbiot_Cell61 or
                                                         p_CellId == nbiot_Cell62 or
                                                         p_CellId == nbiot_Cell63 ))
        {v_SysinfoCombination := c1}
    }
    
    // Initialise BCCH info
>>> Skipped Code <<<

After:

  >>> Skipped Code <<< 
// Initialise NAS parameters
    v_NAS_Param := fl_NBIOT_InitNAS_forNAS_Cells( p_CellId);
    
    // SIB5 is not valid
    if (p_SysinfoCombination == c3 or p_SysinfoCombination == c11) { // @sic R5s170755 sic@
      if (p_CellsOnPLMN == LTE_AllCellsOnSamePLMN and (p_CellId == nbiot_Cell54 or
                                                       p_CellId == nbiot_Cell55 or
                                                       p_CellId == nbiot_Cell56 or  // @sic R5s170511 sic@
                                                       p_CellId == nbiot_Cell57 or
                                                       p_CellId == nbiot_Cell58 or
                                                       p_CellId == nbiot_Cell59 or
                                                       p_CellId == nbiot_Cell60 or
                                                       p_CellId == nbiot_Cell62 or
                                                       p_CellId == nbiot_Cell63 )){    
          if (p_SysinfoCombination == c3){
              v_SysinfoCombination := c1}
          else{
              v_SysinfoCombination := c8} //NTN case WI1233835
      }
      if (p_CellsOnPLMN == LTE_CellsOnDifferentPLMN and (p_CellId == nbiot_Cell50 or
                                                         p_CellId == nbiot_Cell51 or
                                                         p_CellId == nbiot_Cell52 or  // @sic R5s170511 sic@
                                                         p_CellId == nbiot_Cell53 or
                                                         p_CellId == nbiot_Cell55 or
                                                         p_CellId == nbiot_Cell56 or
                                                         p_CellId == nbiot_Cell58 or
                                                         p_CellId == nbiot_Cell61 or
                                                         p_CellId == nbiot_Cell62 or
                                                         p_CellId == nbiot_Cell63 )){
          if (p_SysinfoCombination == c3){
              v_SysinfoCombination := c1}
          else{
              v_SysinfoCombination := c8} //NTN case WI1233835
      }
    }
    
    // Initialise BCCH info
>>> Skipped Code <<<

2.2 Correction to fl_NBIOT_CheckSysinfoCombination
	Function name
	fl_NBIOT_CheckSysinfoCombination

	Reason for change
	System information combination c11 is not defined as the NTN counterpart of c3 in this function. This function handles the case of creation of nbiot_Cell5, nbiot_Cell14, nbiot_Cell12, nbiot_Cell13 and nbiot_Cell18 which cannot have SIB5 assigned to them. As such in any TC that uses c11 as the sys info combination an error on cell creation is raised. This effects the TC 22.4.14 where c3 is used as a combination and is changed to c11 in the init function.



	Summary of change
	Added c11 as a case which will change the sys info combination to c8 which is the NTN counterpart of c1.

	TTCN module
	NBIOT_CellInfo.ttcn

	MCC160 Comment
	




Before:


function fl_NBIOT_CheckSysinfoCombination(NBIOT_CellId_Type p_CellId,
                                            NB_Combination_Type p_SysinfoCombination) return NB_Combination_Type
  {
    var NB_Combination_Type v_SysinfoCombination := p_SysinfoCombination;
    
    // Some cells (14, 12, 13) cannot be assigned to combination c3 as no content is specified for SIB5 in 36.508
    
    select (p_CellId) {
      case (nbiot_Cell5, nbiot_Cell14, nbiot_Cell12, nbiot_Cell13, nbiot_Cell18) {
        select (p_SysinfoCombination) {
          case (c3)  {v_SysinfoCombination := c1}
        }
      }
    }
    return v_SysinfoCombination;
  }

After:


function fl_NBIOT_CheckSysinfoCombination(NBIOT_CellId_Type p_CellId,
                                            NB_Combination_Type p_SysinfoCombination) return NB_Combination_Type
  {
    var NB_Combination_Type v_SysinfoCombination := p_SysinfoCombination;
    
    // Some cells (14, 12, 13) cannot be assigned to combination c3 as no content is specified for SIB5 in 36.508
    
    select (p_CellId) {
      case (nbiot_Cell5, nbiot_Cell14, nbiot_Cell12, nbiot_Cell13, nbiot_Cell18) {
        select (p_SysinfoCombination) {
          case (c3)  {v_SysinfoCombination := c1}
			  case (c11) {v_SysinfoCombination := c8}
        }
      }
    }
    return v_SysinfoCombination;
  }



[bookmark: _Hlk158730384]2.3 Correction to f_NBIOT_Init function

	Function name
	f_NBIOT_Init

	Reason for change
	No case for c6 is present for NTN configuration to convert the system information combination to c11.

	Summary of change
	Added case for c6 to change p_SysinfoCombination to c12.

	TTCN module
	NBIOT_CommonProcedures.ttcn, NBIOT_CellInfo.ttcn

	MCC160 Comment
	



Before:


>>> Skipped Code <<<

  function f_NBIOT_Init(NB_Combination_Type p_SysinfoCombination,
                        IOT_STATE_Type p_AllowAttachWithoutPDN := CONTROL_PLANE,
                        EUTRA_NAS_NumOfPLMN_Type p_CellsOnSamePLMN := LTE_AllCellsOnSamePLMN,
                        EUTRA_NAS_CellSpecialFreq p_CellSpecialFreq := LTE_NAS_NormalFreq,
                        template (omit)NBIOT_CellId_List_Type p_Neighbour_NTN_Cells := omit) runs on NBIOT_PTC
  {
    var NB_InitParams_Local_Type v_Params := {
      CellsOnSamePLMN    := p_CellsOnSamePLMN,
      SysinfoCombination := p_SysinfoCombination,
      CellSpecialFreq    := p_CellSpecialFreq
    };
    var boolean v_DoAttachWithoutPDN := false;
    if (p_AllowAttachWithoutPDN == CONTROL_PLANE) { // @sic R5-171992 sic@
      v_DoAttachWithoutPDN := true;
    }
      
    if(px_ePrimaryFrequencyBand == 255 or px_ePrimaryFrequencyBand == 256) {//@sic R5s230512, R5-235473 sic@
        if (px_NTN_ScenarioUnderTest == GSO){//@sic R5-235437, R5-235473 sic@
            vc_NBIOT_Global.MobileInfo.NTN_SatelliteType := ntn_GSO; //@sic R5-231915, R5-233442 sic@
        }
        else{
            vc_NBIOT_Global.MobileInfo.NTN_SatelliteType := ntn_NGSO
        }
        
         select(p_SysinfoCombination)
        {//@sic  R5-237459, R5-237126 mapping legacy TC SysinfoCombination to NTN-specific SysinfoCombination  sic@
            case(c1)
            {
                v_Params.SysinfoCombination := c8;
            }
            case(c2)
            {
                v_Params.SysinfoCombination := c10;
            }
            case(c3)
            {
                v_Params.SysinfoCombination := c11;
            }
            case(c4)
            {
                v_Params.SysinfoCombination := c13;   //WA#WI=1243268
            }
            case (c8,c9)
            {
            } //WA#WI=1227427 when c8 / c9 is passed no update is required
            case else
            {
                FatalError(__FILE__, __LINE__, "Invalid sysinfo combination");  //no corresponding SysInfo Combination for NTN
            }        }
    }
    else
    {
         vc_NBIOT_Global.MobileInfo.NTN_SatelliteType := ntn_None;
        
    }
    
    fl_NBIOT_Common_Init(v_Params, v_DoAttachWithoutPDN, p_Neighbour_NTN_Cells);   //@sic R5-235468 sic@
  }



After


>>> Skipped Code <<<

  function f_NBIOT_Init(NB_Combination_Type p_SysinfoCombination,
                        IOT_STATE_Type p_AllowAttachWithoutPDN := CONTROL_PLANE,
                        EUTRA_NAS_NumOfPLMN_Type p_CellsOnSamePLMN := LTE_AllCellsOnSamePLMN,
                        EUTRA_NAS_CellSpecialFreq p_CellSpecialFreq := LTE_NAS_NormalFreq,
                        template (omit)NBIOT_CellId_List_Type p_Neighbour_NTN_Cells := omit) runs on NBIOT_PTC
  {
    var NB_InitParams_Local_Type v_Params := {
      CellsOnSamePLMN    := p_CellsOnSamePLMN,
      SysinfoCombination := p_SysinfoCombination,
      CellSpecialFreq    := p_CellSpecialFreq
    };
    var boolean v_DoAttachWithoutPDN := false;
    if (p_AllowAttachWithoutPDN == CONTROL_PLANE) { // @sic R5-171992 sic@
      v_DoAttachWithoutPDN := true;
    }
      
    if(px_ePrimaryFrequencyBand == 255 or px_ePrimaryFrequencyBand == 256) {//@sic R5s230512, R5-235473 sic@
        if (px_NTN_ScenarioUnderTest == GSO){//@sic R5-235437, R5-235473 sic@
            vc_NBIOT_Global.MobileInfo.NTN_SatelliteType := ntn_GSO; //@sic R5-231915, R5-233442 sic@
        }
        else{
            vc_NBIOT_Global.MobileInfo.NTN_SatelliteType := ntn_NGSO
        }
        
         select(p_SysinfoCombination)
        {//@sic  R5-237459, R5-237126 mapping legacy TC SysinfoCombination to NTN-specific SysinfoCombination  sic@
            case(c1)
            {
                v_Params.SysinfoCombination := c8;
            }
            case(c2)
            {
                v_Params.SysinfoCombination := c10;
            }
            case(c3)
            {
                v_Params.SysinfoCombination := c11;
            }
            case(c4)
            {
                v_Params.SysinfoCombination := c13;   //WA#WI=1243268
            }
            case (c6) // WI 1233835
            {
                v_Params.SysinfoCombination := c12
            }
            case (c8,c9)
            {
            } //WA#WI=1227427 when c8 / c9 is passed no update is required
            case else
            {
                FatalError(__FILE__, __LINE__, "Invalid sysinfo combination");  //no corresponding SysInfo Combination for NTN
            }
       }
    }
    else
    {
         vc_NBIOT_Global.MobileInfo.NTN_SatelliteType := ntn_None;
        
    }
    
    fl_NBIOT_Common_Init(v_Params, v_DoAttachWithoutPDN, p_Neighbour_NTN_Cells);   //@sic R5-235468 sic@
  }

2.4 Addition of function f_NBIOT_CellInfo_Get_SI_Index
	Function name
	f_NBIOT_CellInfo_Get_SI_Index

	Reason for change
	In get and set functions of NBIOT_CellInfo.ttcn, hardcoded values are used across many functions which make the code hard to follow and makes any changes to SIB SI location take a long time due to having to change many different functions.

	Summary of change
	Introduced a new function to return index numbers of SIB SI location based on the sys info combination of a cell. The reason for this is to be able to quickly make changes to SI location which and allows for all this to be kept in one place.
Takes CellID, SIB_no (SIB 2 = 2, SIB3 = 3, SIB 5 = 5 etc) and returns an interger which is the index of the SIB in the SI list. This function follows the Scheduling information No. in 36.508 - Tables 8.1.4.3.1.2.

	TTCN module
	NBIOT_CellInfo.ttcn

	MCC160 Comment
	



   /*
   * @desc     returns the index of the specified SIB in the SI list given the cell ID and SIB no
   * @param    p_CellId
   * @param    p_SIB_no
   * @status   Unapproved, Testing
   */
  // WI 1233835  
  function f_NBIOT_CellInfo_Get_SI_Index(NBIOT_CellId_Type p_CellId,
                                         integer p_SIB_no) runs on NBIOT_PTC return integer
    {
      var template (value) NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
      var NB_Combination_Type v_Sys_Comb := v_CellInfo.Sysinfo.Combination;
      select (p_SIB_no){
          case(2){
              return 0
          }
          case(3){
              select (v_Sys_Comb){
                  case(c2, c3, c4, c9){
                      return 1
                  }
                  case(c10, c11,c13){
                      return 2    
                  }
                  case else{
                      FatalError(__FILE__, __LINE__, "Invalid system combination for SIB3");
                  }
                  
              }
          }
          case(4){
              select(v_Sys_Comb){
                  case(c2){
                      return 2
                  }
                  case(c10){
                      return 3
                  }
                  case else{
                      FatalError(__FILE__, __LINE__, "Invalid system combination for SIB4");
                  }
              }
          }
          case(5){
              select(v_Sys_Comb){
                  case(c3){
                      return 2
                  }
                  case(c11){
                      return 3
                  }
                  case else{
                      FatalError(__FILE__, __LINE__, "Invalid system combination for SIB5");
                  }
              }  
          }
          case(14){
              select(v_Sys_Comb){
                  case(c4){
                      return 2
                  }
                  case(c13){
                      return 3
                  }
                  case else{
                      FatalError(__FILE__, __LINE__, "Invalid system combination for SIB41");
                  }
              }  
          }
          //case (16){} //standin
          case(20){
              return 1
          }
          case (22){
              select(v_Sys_Comb){
                  case(c6, c7){
                      return 1
                  }
                  case(c12){
                      return 2
                  }
                  case else{
                      FatalError(__FILE__, __LINE__, "Invalid system combination for SIB22");
                  }
              }
          }
          case (23){
              return 2    
          }
          case(31){
              select(v_Sys_Comb){
                  case(c9){
                      return 0    
                  }
                  case(c8, c10, c11, c12, c13){
                      return 1
                  }
                  case else{
                      FatalError(__FILE__, __LINE__, "Invalid system combination for SIB31");
                  }
              }
          }
          //case(32)//standin
          case else{
              FatalError(__FILE__, __LINE__, "Invalid System Block Number\System Information Combination");
          }
      }
    }
2.5 Addition of function f_NBIOT_Init_SIBs
	Function name
	f_NBIOT_Init_SIBs

	Reason for change
	f_NBIOT_InitSystemInformation is cluttered and had a hard to read control flow which made it hard to read and edit. A large part of this is making the SI list which holds the SIBs in a specific order with certain initialisations needed. 

	Summary of change
	To help readability and maintainability this function takes an interger for the SIB number (SIB2 = 2, SIB3 = 3, SIB5 = 5), the SI list to build around as an inout variable, and if needed (for certain SIB initialisations) CellId, FrequnecyBand, Opertation Mode and CellsOnPLMN. No value is returned as the function is paramatized with the use of the inout parameter p_SI_List.
Depending on the SIB number passed, that case statement is selected, and the SIB is initialised and added to the SI list.
The SI list is built one SIB at a time with each call adding that SIB on to the end of the list (append nature).
The individual SIB initialisation is done the same as the original NBIOT_InitSystemInformation. (Spec 36.508 - 8.1.4.3.1)

	TTCN module
	NBIOT_CellInfo.ttcn

	MCC160 Comment
	



/*
   * @desc      function initialising SIBs for NBIOT as per 36.508 (NOTE when calling multiple times ensure that each referance p_BlockTypeNo is bigger than the last call to ensure proper layout of v_SI_List)
   * @params    p_BlockTypeNo
   * @params    p_SI_List (passes by refereance)
   * @params    p_CellId (omit)
   * @params    p_FrequencyBand (omit)
   * @params    p_OperationMode (omit_
   * @params    p_CellsOnPLMN (omit)
   * @return    void
   * @status    ??? (NBIOT)
   */
    // Addition of this function is to help readability of f_NBIOT_SystemInformation and to help with addition of new blocks and combinations WI1233835
  function f_NBIOT_Init_SIBs(integer                                       p_BlockTypeNo,
                             inout template (value) NB_SI_List_Type        p_SI_List,
                             template (omit) NBIOT_CellId_Type             p_CellId := omit,
                             template (omit) integer                       p_FrequencyBand := omit,
                             template (omit) NB_OperationMode_Type         p_OperationMode := omit,
                             template (omit) EUTRA_NAS_NumOfPLMN_Type      p_CellsOnPLMN := omit)
    runs on NBIOT_PTC {
      var template (value) SystemInformation_NB_r13_IEs v_SIB_SIXX;
      var integer p_SI_List_Index := lengthof(p_SI_List);
      var NBIOT_CellId_Type v_CellId;
      var integer v_FrequencyBand;
      var NB_OperationMode_Type v_OperationMode;
      var EUTRA_NAS_NumOfPLMN_Type v_CellsOnPLMN;
 
      select (p_BlockTypeNo) { // blocktype to initialise
        case (2) { // BlockType2
            if (ispresent(p_CellId) and ispresent(p_FrequencyBand) and ispresent(p_OperationMode) and ispresent(p_CellsOnPLMN)){
                v_CellId := valueof(p_CellId);
                v_FrequencyBand := valueof(p_FrequencyBand);
                v_OperationMode := valueof(p_OperationMode);
                v_CellsOnPLMN := valueof(p_CellsOnPLMN);
                 
                v_SIB_SIXX := cs_SI_NB_SIB2(f_NBIOT_InitSystemInformation2(v_CellId, v_FrequencyBand, v_OperationMode, v_CellsOnPLMN));
                p_SI_List[0] := cs_SystemInformation_Common_NB(v_SIB_SIXX); // SIB2 always first on p_SI_List
            }
            else {
                FatalError(__FILE__, __LINE__, "Not enough parameters were passed to f_NBIOT_Init_SIBs");
            }
        }
        case (3) { // BlockType3
            v_SIB_SIXX := cs_SI_NB_SIB3(cs_508_SystemInformationBlockType3_NB_Def);
            p_SI_List[p_SI_List_Index] := cs_SystemInformation_Common_NB(v_SIB_SIXX);
        }
        case (4) { // BlockType4
            v_SIB_SIXX := cs_SI_NB_SIB4(cs_508_SystemInformationBlockType4_NB_Def);
            p_SI_List[p_SI_List_Index] := cs_SystemInformation_Common_NB(v_SIB_SIXX);
        }
        case (5) { // BlockType5
            if (ispresent(p_CellId) and ispresent(p_FrequencyBand) and ispresent(p_OperationMode)){
                v_CellId := valueof(p_CellId);
                v_FrequencyBand := valueof(p_FrequencyBand);
                v_OperationMode := valueof(p_OperationMode);
 
                v_SIB_SIXX := cs_SI_NB_SIB5(f_NBIOT_InitSystemInformation5(v_CellId, v_FrequencyBand, v_OperationMode));
                p_SI_List[p_SI_List_Index] := cs_SystemInformation_Common_NB(v_SIB_SIXX);
            }
            else {
                FatalError(__FILE__, __LINE__, "Not enough parameters were passed to f_NBIOT_Init_SIBs");
            }
        }
        case (14) { // BlockType14
            v_SIB_SIXX := cs_SI_NB_SIB14(cs_SystemInformationBlockType14_NB_Def);
            p_SI_List[p_SI_List_Index] := cs_SystemInformation_Common_NB(v_SIB_SIXX);
        }
        case (20) { // BlockType20
            v_SIB_SIXX := cs_SI_NB_SIB20(cs_508_SystemInformationBlockType20_NB_Def);
            p_SI_List[p_SI_List_Index] := cs_SystemInformation_Common_NB(v_SIB_SIXX);
        }
        case (22) { // BlockType22
            if (ispresent(p_CellId) and ispresent(p_FrequencyBand) and ispresent(p_OperationMode) and ispresent(p_CellsOnPLMN)){
                v_CellId := valueof(p_CellId);
                v_FrequencyBand := valueof(p_FrequencyBand);
                v_OperationMode := valueof(p_OperationMode);
                v_CellsOnPLMN := valueof(p_CellsOnPLMN);
                 
                 
                var NB_CarrierFreq_Type v_CarrierFreq := fl_NBIOT_Get_CarrierFreq(v_CellId, v_FrequencyBand, v_OperationMode, v_CellsOnPLMN); // @sic R5-185113 sic@
                v_SIB_SIXX := cs_SI_NB_SIB22(cs_508_SystemInformationBlockType22_NB_Def(cs_DL_CarrierConfigCommon_NB_r14(v_CarrierFreq.DL.carrierFreq_r13, v_CarrierFreq.DL.carrierFreqOffset_r13), v_CarrierFreq.UL));
                p_SI_List[p_SI_List_Index] := cs_SystemInformation_Common_NB(v_SIB_SIXX);
            }
            else {
                FatalError(__FILE__, __LINE__, "Not enough parameters were passed to f_NBIOT_Init_SIBs");
            }
        }
        case (23) { // BlockType23
            v_SIB_SIXX := cs_SI_NB_SIB23(cs_508_SystemInformationBlockType23_NB_Def);
            p_SI_List[p_SI_List_Index] := cs_SystemInformation_Common_NB(v_SIB_SIXX);
        }
        case (31) { // BlockType31 // Adds SIB32 and bundles with SIB2 WI1182332
            p_SI_List[p_SI_List_Index] := cs_SystemInformation_Common_NB(f_NBIOT_InitSIB31());
            //p_SI_List[0] := cs_SystemInformation_Common_NB(cs_SI_NB_SIB2_and_SIB31(v_SIB2.sib_TypeAndInfo_r13[0].sib2_r13, f_NBIOT_InitSIB31().sib_TypeAndInfo_r13[0].sib31_v1700)); // @sic R5-235468 sic@
        }
        case (32) {
            FatalError(__FILE__, __LINE__, "System block type 32 not yet supported")
        }  
        case else { // BlockType not in supported Blocktypes
            FatalError(__FILE__, __LINE__, "Block no. , " & int2str(p_BlockTypeNo) & ", not supported in this init function");
        }
      }
  }
2.6 Correction to f_NBIOT_InitSystemInformation
	Function name
	f_NBIOT_InitSystemInformation

	Reason for change
	f_NBIOT_InitSystemInformation is cluttered and had a hard to read control flow which made it hard to read and edit. A large part of this is making the SI list which holds the SIBs in a specific order with certain initialisations needed. 

	Summary of change
	To help readability and maintainability this function has been changed to use select and case statements for each sys info combination possibility. 
This change divides all current 13 combinations into separate blocks for initialisation. This in combination with the new function above in [2.5] f_NBIOT_Init_SIBs helps readability and maintainability. 
Each control flow path from the original function has been reproduced in each case.
(Spec 36.508)


	TTCN module
	NBIOT_CellInfo.ttcn

	MCC160 Comment
	



Before:

        function f_NBIOT_InitSystemInformation(NBIOT_CellId_Type                         p_CellId,
                                         EUTRA_ASN1_CellIdentity_Type              p_CellIdentity,
                                         NB_Combination_Type                       p_Combination,
                                         integer                                   p_FrequencyBand,
                                         NB_OperationMode_Type                     p_OperationMode,
                                         template (value) PLMN_IdentityList_NB_r13 p_PLMN_Identity,
                                         EUTRA_ASN1_TrackingAreaCode_Type          p_TrackingAreaCode,
                                         EUTRA_NAS_NumOfPLMN_Type p_CellsOnPLMN := LTE_AllCellsOnSamePLMN)
     runs on NBIOT_PTC return template (value) NB_BcchInfo_Type
  {
    var template (value) NB_BcchInfo_Type v_CellBcchInfo;
    var template (value) BCCH_BCH_Message_NB v_MIB;
    var template (value) BCCH_DL_SCH_Message_NB v_SIB1;
    var template (value) NB_SI_List_Type v_SI_List;
    var template (value) SystemInformation_NB_r13_IEs v_SIB2;
    var template (value) SystemInformation_NB_r13_IEs v_SIB_SI2;
    var template (value) SystemInformation_NB_r13_IEs v_SIB_SI3;
    var template (omit) NRS_CRS_PowerOffset_Type v_NRS_CRS_PowerOffset;
    var template (omit) NB_EutraControlRegionSize_Type v_NB_EutraControlRegionSize;
    var template (value) NB_OperationModeInfo_Type v_OperationMode;
    var template (value) SchedulingInfoList_NB_r13 v_SchedulingInfoList;
    var SIB_MappingInfo_NB_r13 v_SIB_MappingInfo_SI2;
    var SIB_MappingInfo_NB_r13 v_SIB_MappingInfo_SI3;
    var NB_CarrierFreq_Type v_CarrierFreq := fl_NBIOT_Get_CarrierFreq(p_CellId, p_FrequencyBand, p_OperationMode, p_CellsOnPLMN); // @sic R5-185113 sic@

    select (p_OperationMode) {
      case (standalone) {
        v_OperationMode := cs_NB_OperationModeInfo_Standalone;
        v_NB_EutraControlRegionSize := omit;
        v_NRS_CRS_PowerOffset := omit;
      }
      case (inband_samePCI) {
        v_OperationMode := cs_NB_OperationModeInfo_InBandSamePCI (fl_NBIOT_MIB_GetEUTRA_CRS_SequenceInfo(p_FrequencyBand)); // @sic R5-181295 sic@
        v_NB_EutraControlRegionSize := n2;
        v_NRS_CRS_PowerOffset := dB0;
      }
      case (inband_differentPCI) {
        v_OperationMode := cs_NB_OperationModeInfo_InBandDifferentPCI (fl_NBIOT_MIB_GetRasterOffset (p_CellId, p_FrequencyBand, p_OperationMode, p_CellsOnPLMN)); // @sic R5-176998 R5s180571 sic@
        v_NB_EutraControlRegionSize := n2;
        v_NRS_CRS_PowerOffset := omit;
      }
      case (guardband) {
        v_OperationMode := cs_NB_OperationModeInfo_GuardBand (fl_NBIOT_MIB_GetRasterOffset (p_CellId, p_FrequencyBand, p_OperationMode, p_CellsOnPLMN)); // @sic R5-176998 R5s180571 sic@
        v_NB_EutraControlRegionSize := omit;
        v_NRS_CRS_PowerOffset := omit;
      }
    }
    // --- Initialise MIB ---
    v_MIB :=  cs_508_MasterInformationBlock_NB_Init(v_OperationMode);
    
    // --- Initialise SIB1 ---
    // -- A default scheduling assigned ----------
    v_SIB1 := cs_508_SystemInformationBlockType1_NB_Def(p_PLMN_Identity,
                                                        p_TrackingAreaCode,
                                                        p_CellIdentity,
                                                        p_FrequencyBand,
                                                        v_NB_EutraControlRegionSize,
                                                        v_NRS_CRS_PowerOffset,
                                                        cs_SchedulingInfo_NB_1SI);
    // --- Initialise SIB2 ---
    v_SIB2 := cs_SI_NB_SIB2(f_NBIOT_InitSystemInformation2(p_CellId, p_FrequencyBand, p_OperationMode, p_CellsOnPLMN));
    v_SI_List[0] := cs_SystemInformation_Common_NB(v_SIB2);
    
    // --- Initialise all SYSINFO --
    if (p_Combination == c1) {
      v_SchedulingInfoList := cs_SchedulingInfo_NB_1SI;
    } else if (p_Combination == c5) { // @sic R5-185113 sic@
      v_SI_List[1] := cs_SystemInformation_Common_NB(cs_SI_NB_SIB20(cs_508_SystemInformationBlockType20_NB_Def));
      v_SchedulingInfoList := cs_SchedulingInfo_NB_2SI({sibType20_NB_r14});
    } else if (p_Combination == c6) { // @sic R5-185113 sic@
      v_SI_List[1] := cs_SystemInformation_Common_NB(cs_SI_NB_SIB22(cs_508_SystemInformationBlockType22_NB_Def(cs_DL_CarrierConfigCommon_NB_r14(v_CarrierFreq.DL.carrierFreq_r13, v_CarrierFreq.DL.carrierFreqOffset_r13), v_CarrierFreq.UL)));
      v_SchedulingInfoList := cs_SchedulingInfo_NB_2SI({sibType22_NB_r14});
    } else if (p_Combination == c7) { //@sic R5-2226052 sic@
      v_SI_List[1] := cs_SystemInformation_Common_NB(cs_SI_NB_SIB22(cs_508_SystemInformationBlockType22_NB_Def(cs_DL_CarrierConfigCommon_NB_r14(v_CarrierFreq.DL.carrierFreq_r13, v_CarrierFreq.DL.carrierFreqOffset_r13), v_CarrierFreq.UL)));
      v_SI_List[2] := cs_SystemInformation_Common_NB(cs_SI_NB_SIB23(cs_508_SystemInformationBlockType23_NB_Def));
      v_SchedulingInfoList := cs_SchedulingInfo_NB_3SI(-, -, -, {sibType22_NB_r14}, -, -, -, {});
      v_SIB1.message_.c1.systemInformationBlockType1_r13.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.schedulingInfoList_v1530 := cs_SchedulingInfo_NB_c7_r15;
    } else if (p_Combination == c8 or p_Combination == c10 or p_Combination == c11 or p_Combination == c12 or p_Combination == c13) { //@sic R5-231904, R5-231933, R5-233367, R5-233365, R5-235331, R5s230706 sic@
        v_SI_List[1] := cs_SystemInformation_Common_NB(f_NBIOT_InitSIB31()); //@sic R5-235420, R5-235468 sic@
        v_SchedulingInfoList := cs_SchedulingInfo_NB_2SI({});
        v_MIB.message_.partEARFCN_17.earfcn_LSB := fl_NBIOT_Get_EARFCN_2LSB(v_CarrierFreq);
        v_MIB.message_.schedulingInfoSIB1_r13 := 5; //@sic R5-235420 sic@
        v_SIB1.message_.c1.systemInformationBlockType1_r13.cellAccessRelatedInfo_r13.cellBarred_r13 := barred;
        v_SIB2.sib_TypeAndInfo_r13[0].sib2_r13.radioResourceConfigCommon_r13.ntn_ConfigCommon_r17 := cs_508_NTN_ConfigCommon_r17;
        v_SI_List[0] := cs_SystemInformation_Common_NB(v_SIB2); //@sic R5s230478 sic@
        v_SIB1.message_.c1.systemInformationBlockType1_r13.nonCriticalExtension := cs_SystemInformationBlockType1_NB_v1350_NTN_r17(p_TrackingAreaCode); //@sic R5s230706 sic@
        if (p_Combination == c10 or p_Combination == c11 or p_Combination == c12 or p_Combination == c13){ //@sic R5s230706 Ch1 sic@
            if(p_Combination == c10 or p_Combination == c11 or p_Combination == c13){ //c8+SIB3
                v_SI_List[2] := cs_SystemInformation_Common_NB(cs_SI_NB_SIB3(cs_508_SystemInformationBlockType3_NB_Def));
                if(p_Combination == c10){ //+ SIB4
                    v_SI_List[3] := cs_SystemInformation_Common_NB(cs_SI_NB_SIB4(cs_508_SystemInformationBlockType4_NB_Def));
                    v_SchedulingInfoList := cs_SchedulingInfo_NB_4SI_NTN({sibType4_NB_r13});
                }
                else if(p_Combination == c11){ //+ SIB5
                    v_SI_List[3] := cs_SystemInformation_Common_NB(cs_SI_NB_SIB5(f_NBIOT_InitSystemInformation5(p_CellId, p_FrequencyBand, p_OperationMode)));
                    v_SchedulingInfoList := cs_SchedulingInfo_NB_4SI_NTN({sibType5_NB_r13});
                }
                else{//+ SIB14
                    v_SI_List[3] := cs_SystemInformation_Common_NB(cs_SI_NB_SIB14(cs_SystemInformationBlockType14_NB_Def));
                    v_SchedulingInfoList := cs_SchedulingInfo_NB_4SI_NTN({sibType14_NB_r13});
                }
                v_SIB1.message_.c1.systemInformationBlockType1_r13.nonCriticalExtension := cs_SystemInformationBlockType1_NB_v1350_NTN_r17(p_TrackingAreaCode, cs_SchedulingInfo_NB_c10_c11_c13_r15);
            }
            else{//c12 = c8 + SIB22
                v_SI_List[2] := cs_SystemInformation_Common_NB(cs_SI_NB_SIB22(cs_508_SystemInformationBlockType22_NB_Def(cs_DL_CarrierConfigCommon_NB_r14(v_CarrierFreq.DL.carrierFreq_r13, v_CarrierFreq.DL.carrierFreqOffset_r13), v_CarrierFreq.UL)));
                v_SchedulingInfoList := cs_SchedulingInfo_NB_3SI(-, -, -, {}, -, -, -, {sibType22_NB_r14});
                v_SIB1.message_.c1.systemInformationBlockType1_r13.nonCriticalExtension := cs_SystemInformationBlockType1_NB_v1350_NTN_r17(p_TrackingAreaCode, cs_SchedulingInfo_NB_c12_r15);
            }
        }
    } else if (p_Combination == c9)  {//@sic R5-235420, R5s230706 sic@
            v_SIB2.sib_TypeAndInfo_r13[0].sib2_r13.radioResourceConfigCommon_r13.ntn_ConfigCommon_r17 := cs_508_NTN_ConfigCommon_r17;
            v_SI_List[0] :=cs_SystemInformation_Common_NB(cs_SI_NB_SIB2_and_SIB31(v_SIB2.sib_TypeAndInfo_r13[0].sib2_r13, f_NBIOT_InitSIB31().sib_TypeAndInfo_r13[0].sib31_v1700)); // @sic R5-235468 sic@
            v_SI_List[1] := cs_SystemInformation_Common_NB(cs_SI_NB_SIB3(cs_508_SystemInformationBlockType3_NB_Def));
            v_SchedulingInfoList := cs_SchedulingInfo_NB_2SI({sibType3_NB_r13});
            v_SIB1.message_.c1.systemInformationBlockType1_r13.nonCriticalExtension := cs_SystemInformationBlockType1_NB_v1350_NTN_r17(p_TrackingAreaCode, cs_SchedulingInfo_NB_c9_r15);
            v_SIB1.message_.c1.systemInformationBlockType1_r13.cellAccessRelatedInfo_r13.cellBarred_r13 := barred; //@sic R5s230706 Ch1 sic@
            v_MIB.message_.partEARFCN_17.earfcn_LSB := fl_NBIOT_Get_EARFCN_2LSB(v_CarrierFreq);
            v_MIB.message_.schedulingInfoSIB1_r13 := 5;
            
    } else {
      // --- Initialise SIB3 ---
      v_SIB_MappingInfo_SI2 := {sibType3_NB_r13};
      v_SIB_SI2 := cs_SI_NB_SIB3(cs_508_SystemInformationBlockType3_NB_Def);
      
      select (p_Combination) { // @sic R5-185113 sic@
        case (c2) { // SIB3 & SIB4 Intra-Frequency
          v_SIB_MappingInfo_SI3 := {sibType4_NB_r13};
          v_SIB_SI3 := cs_SI_NB_SIB4(cs_508_SystemInformationBlockType4_NB_Def);
        }
        case (c3) { // SIB 3 & SIB5 Inter-Frequency
          v_SIB_MappingInfo_SI3 := {sibType5_NB_r13};
          v_SIB_SI3 := cs_SI_NB_SIB5(f_NBIOT_InitSystemInformation5(p_CellId, p_FrequencyBand, p_OperationMode));
        }
        case (c4) { // SIB 3 & 14
          v_SIB_MappingInfo_SI3 := {sibType14_NB_r13};
          v_SIB_SI3 := cs_SI_NB_SIB14(cs_SystemInformationBlockType14_NB_Def);
        }
        case else {
          FatalError(__FILE__, __LINE__, "invalid sysinfo combination");
        }
      }
      v_SI_List[1] := cs_SystemInformation_Common_NB(v_SIB_SI2);
      v_SI_List[2] := cs_SystemInformation_Common_NB(v_SIB_SI3);
      v_SchedulingInfoList := cs_SchedulingInfo_NB_3SI(-, -, -, v_SIB_MappingInfo_SI2, -, -, -, v_SIB_MappingInfo_SI3);
    }
    v_SIB1.message_.c1.systemInformationBlockType1_r13.schedulingInfoList_r13 := v_SchedulingInfoList;
    v_CellBcchInfo := cs_BCCHInfo_NB_C1(v_MIB, v_SIB1, v_SI_List);
    return v_CellBcchInfo;
  }

After:

          function f_NBIOT_InitSystemInformation(NBIOT_CellId_Type                         p_CellId,
                                         EUTRA_ASN1_CellIdentity_Type              p_CellIdentity,
                                         NB_Combination_Type                       p_Combination,
                                         integer                                   p_FrequencyBand,
                                         NB_OperationMode_Type                     p_OperationMode,
                                         template (value) PLMN_IdentityList_NB_r13 p_PLMN_Identity,
                                         EUTRA_ASN1_TrackingAreaCode_Type          p_TrackingAreaCode,
                                         EUTRA_NAS_NumOfPLMN_Type p_CellsOnPLMN := LTE_AllCellsOnSamePLMN)
     runs on NBIOT_PTC return template (value) NB_BcchInfo_Type
  {
    var template (value) NB_BcchInfo_Type v_CellBcchInfo;
    var template (value) BCCH_BCH_Message_NB v_MIB;
    var template (value) BCCH_DL_SCH_Message_NB v_SIB1;
    var template (value) NB_SI_List_Type v_SI_List;
    var template (value) SystemInformation_NB_r13_IEs v_SIB2;
    var template (omit) NRS_CRS_PowerOffset_Type v_NRS_CRS_PowerOffset;
    var template (omit) NB_EutraControlRegionSize_Type v_NB_EutraControlRegionSize;
    var template (value) NB_OperationModeInfo_Type v_OperationMode;
    var template (value) SchedulingInfoList_NB_r13 v_SchedulingInfoList;
    var NB_CarrierFreq_Type v_CarrierFreq := fl_NBIOT_Get_CarrierFreq(p_CellId, p_FrequencyBand, p_OperationMode, p_CellsOnPLMN); // @sic R5-185113 sic@

    select (p_OperationMode) {
      case (standalone) {
        v_OperationMode := cs_NB_OperationModeInfo_Standalone;
        v_NB_EutraControlRegionSize := omit;
        v_NRS_CRS_PowerOffset := omit;
      }
      case (inband_samePCI) {
        v_OperationMode := cs_NB_OperationModeInfo_InBandSamePCI (fl_NBIOT_MIB_GetEUTRA_CRS_SequenceInfo(p_FrequencyBand)); // @sic R5-181295 sic@
        v_NB_EutraControlRegionSize := n2;
        v_NRS_CRS_PowerOffset := dB0;
      }
      case (inband_differentPCI) {
        v_OperationMode := cs_NB_OperationModeInfo_InBandDifferentPCI (fl_NBIOT_MIB_GetRasterOffset (p_CellId, p_FrequencyBand, p_OperationMode, p_CellsOnPLMN)); // @sic R5-176998 R5s180571 sic@
        v_NB_EutraControlRegionSize := n2;
        v_NRS_CRS_PowerOffset := omit;
      }
      case (guardband) {
        v_OperationMode := cs_NB_OperationModeInfo_GuardBand (fl_NBIOT_MIB_GetRasterOffset (p_CellId, p_FrequencyBand, p_OperationMode, p_CellsOnPLMN)); // @sic R5-176998 R5s180571 sic@
        v_NB_EutraControlRegionSize := omit;
        v_NRS_CRS_PowerOffset := omit;
      }
    }
    // --- Initialise MIB ---
    v_MIB :=  cs_508_MasterInformationBlock_NB_Init(v_OperationMode);
    
    // --- Initialise SIB1 ---
    // -- A default scheduling assigned ----------
    v_SIB1 := cs_508_SystemInformationBlockType1_NB_Def(p_PLMN_Identity,
                                                        p_TrackingAreaCode,
                                                        p_CellIdentity,
                                                        p_FrequencyBand,
                                                        v_NB_EutraControlRegionSize,
                                                        v_NRS_CRS_PowerOffset,
                                                        cs_SchedulingInfo_NB_1SI);
    // --- Initialise SIB2 --- 
    v_SIB2 := cs_SI_NB_SIB2(f_NBIOT_InitSystemInformation2(p_CellId, p_FrequencyBand, p_OperationMode, p_CellsOnPLMN));
    v_SI_List[0] := cs_SystemInformation_Common_NB(v_SIB2);
    
    // --- Initialise all SYSINFO --
    select(p_Combination) { // re-design of Combination selecting and addition of combinations c10, c11 and c12 WI1233835
        case (c1){
            // construct the scheduling info list
            v_SchedulingInfoList := cs_SchedulingInfo_NB_1SI;
        }
        case (c2){ // SIB 2, 3, 4
            // Add SIBs to v_SI_List
            f_NBIOT_Init_SIBs(3, v_SI_List);
            f_NBIOT_Init_SIBs(4, v_SI_List);
            
            // construct the scheduling info list
            v_SchedulingInfoList := cs_SchedulingInfo_NB_3SI(-, -, -, {sibType3_NB_r13}, -, -, -, {sibType4_NB_r13}); 
        }
        case (c3){ // SIB 2, 3, 5
            // Add SIBs to v_SI_List
            f_NBIOT_Init_SIBs(3, v_SI_List);
            f_NBIOT_Init_SIBs(5, v_SI_List, p_CellId, p_FrequencyBand, p_OperationMode);
            
            // construct the scheduling info list
            v_SchedulingInfoList := cs_SchedulingInfo_NB_3SI(-, -, -, {sibType3_NB_r13}, -, -, -, {sibType5_NB_r13});
        }
        case (c4){ // SIB 2, 3, 14
            // Add SIBs to v_SI_List
            f_NBIOT_Init_SIBs(3, v_SI_List);
            f_NBIOT_Init_SIBs(14, v_SI_List);
            
            // construct the scheduling info list
            v_SchedulingInfoList := cs_SchedulingInfo_NB_3SI(-, -, -, {sibType3_NB_r13}, -, -, -, {sibType14_NB_r13});
        }
        case (c5){ // SIB 2, 20
            // Add SIBs to v_SI_List
            f_NBIOT_Init_SIBs(20, v_SI_List);
            
            // construct the scheduling info list
            v_SchedulingInfoList := cs_SchedulingInfo_NB_2SI({sibType20_NB_r14});
        }
        case (c6){ // SIB 2, 22
            // Add SIBs to v_SI_List
            f_NBIOT_Init_SIBs(22, v_SI_List, p_CellId, p_FrequencyBand, p_OperationMode, p_CellsOnPLMN);
            
            // construct the scheduling info list
            v_SchedulingInfoList := cs_SchedulingInfo_NB_2SI({sibType22_NB_r14});
        }
        case (c7){ // SIB 2, 22, 23
            // Add SIBs to v_SI_List
            f_NBIOT_Init_SIBs(22, v_SI_List, p_CellId, p_FrequencyBand, p_OperationMode, p_CellsOnPLMN);
            f_NBIOT_Init_SIBs(23, v_SI_List);
            
            // construct the scheduling info list
            v_SchedulingInfoList := cs_SchedulingInfo_NB_3SI(-, -, -, {sibType22_NB_r14}, -, -, -, {});
            v_SIB1.message_.c1.systemInformationBlockType1_r13.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.schedulingInfoList_v1530 := cs_SchedulingInfo_NB_c7_r15;
        }
        case (c8){ // SIB 2, 31
            // Add SIBs to v_SI_List
            f_NBIOT_Init_SIBs(31, v_SI_List); // Sets SIB31 at end of v_SI_List
            
            // NTN SIB31 specific config
            v_MIB.message_.partEARFCN_17.earfcn_LSB := fl_NBIOT_Get_EARFCN_2LSB(v_CarrierFreq);
            v_MIB.message_.schedulingInfoSIB1_r13 := 5; //@sic R5-235420 sic@
            v_SIB1.message_.c1.systemInformationBlockType1_r13.cellAccessRelatedInfo_r13.cellBarred_r13 := barred;
            v_SIB1.message_.c1.systemInformationBlockType1_r13.nonCriticalExtension := cs_SystemInformationBlockType1_NB_v1350_NTN_r17(p_TrackingAreaCode); ////WA#WI=1195606 moved code here to update for c8
            v_SIB2.sib_TypeAndInfo_r13[0].sib2_r13.radioResourceConfigCommon_r13.ntn_ConfigCommon_r17 := cs_508_NTN_ConfigCommon_r17;
            v_SI_List[0] := cs_SystemInformation_Common_NB(v_SIB2); //@sic R5s230478 sic@
            
            //v_MIB.message_.schedulingInfoSIB1_r13 := 5; // MIB value changes to 5 based on 36.213 Table 16.4.1.5.2-1 WA#WI=1172359 MIB value updated to increase the TBS size R5s230550
            
            // construct the scheduling info list
            v_SchedulingInfoList := cs_SchedulingInfo_NB_2SI({});
        }
        case (c9){ // SIB 2, 3, 31
            // Add SIBs to v_SI_List
            f_NBIOT_Init_SIBs(3, v_SI_List);
            //v_SI_List := f_NBIOT_Init_SIBs(31, v_SI_List);
            // SIB 2 and 31 is conjoined in c9
            v_SI_List[0] :=cs_SystemInformation_Common_NB(cs_SI_NB_SIB2_and_SIB31(v_SIB2.sib_TypeAndInfo_r13[0].sib2_r13, f_NBIOT_InitSIB31().sib_TypeAndInfo_r13[0].sib31_v1700)); // @sic R5-235468 sic@

            
            // NTN SIB31 specific config
            v_SIB1.message_.c1.systemInformationBlockType1_r13.cellAccessRelatedInfo_r13.cellBarred_r13 := barred;
            v_SIB2.sib_TypeAndInfo_r13[0].sib2_r13.radioResourceConfigCommon_r13.ntn_ConfigCommon_r17 := cs_508_NTN_ConfigCommon_r17;
            v_SchedulingInfoList := cs_SchedulingInfo_NB_2SI({sibType3_NB_r13});
            v_SIB1.message_.c1.systemInformationBlockType1_r13.nonCriticalExtension := cs_SystemInformationBlockType1_NB_v1350_NTN_r17(p_TrackingAreaCode, cs_SchedulingInfo_NB_c9_r15);
            v_MIB.message_.partEARFCN_17.earfcn_LSB := fl_NBIOT_Get_EARFCN_2LSB(v_CarrierFreq);
            v_MIB.message_.schedulingInfoSIB1_r13 := 5;
            
            // construct the scheduling info list
            v_SchedulingInfoList := cs_SchedulingInfo_NB_2SI({sibType3_NB_r13}); // 2SI as SIB 31 bundled with SIB 2
        }
        case (c10){ // new // SIB 2, 3, 4, 31
            // Add SIBs to v_SI_List
            f_NBIOT_Init_SIBs(31, v_SI_List);
            f_NBIOT_Init_SIBs(3, v_SI_List);
            f_NBIOT_Init_SIBs(4, v_SI_List);
            
            // NTN SIB31 specific config
            v_MIB.message_.partEARFCN_17.earfcn_LSB := fl_NBIOT_Get_EARFCN_2LSB(v_CarrierFreq);
            v_MIB.message_.schedulingInfoSIB1_r13 := 5; //@sic R5-235420 sic@
            v_SIB1.message_.c1.systemInformationBlockType1_r13.cellAccessRelatedInfo_r13.cellBarred_r13 := barred;
            v_SIB1.message_.c1.systemInformationBlockType1_r13.nonCriticalExtension := cs_SystemInformationBlockType1_NB_v1350_NTN_r17(p_TrackingAreaCode); 
            v_SIB2.sib_TypeAndInfo_r13[0].sib2_r13.radioResourceConfigCommon_r13.ntn_ConfigCommon_r17 := cs_508_NTN_ConfigCommon_r17;
            v_SI_List[0] := cs_SystemInformation_Common_NB(v_SIB2); //@sic R5s230478 sic@
            
            
            // construct the scheduling info list
            v_SchedulingInfoList := cs_SchedulingInfo_NB_4SI_NTN({sibType4_NB_r13});
            v_SIB1.message_.c1.systemInformationBlockType1_r13.nonCriticalExtension := cs_SystemInformationBlockType1_NB_v1350_NTN_r17(p_TrackingAreaCode, cs_SchedulingInfo_NB_c10_c11_c13_r15);
        }
        case (c11) { // new // SIB 2, 3, 5, 31
            // Add SIBs to v_SI_List
            f_NBIOT_Init_SIBs(31, v_SI_List);
            f_NBIOT_Init_SIBs(3, v_SI_List);
            f_NBIOT_Init_SIBs(5, v_SI_List, p_CellId, p_FrequencyBand, p_OperationMode);
            
            // NTN SIB31 specific config
            v_MIB.message_.partEARFCN_17.earfcn_LSB := fl_NBIOT_Get_EARFCN_2LSB(v_CarrierFreq);
            v_MIB.message_.schedulingInfoSIB1_r13 := 5; //@sic R5-235420 sic@
            v_SIB1.message_.c1.systemInformationBlockType1_r13.cellAccessRelatedInfo_r13.cellBarred_r13 := barred;
            v_SIB2.sib_TypeAndInfo_r13[0].sib2_r13.radioResourceConfigCommon_r13.ntn_ConfigCommon_r17 := cs_508_NTN_ConfigCommon_r17;
            v_SI_List[0] := cs_SystemInformation_Common_NB(v_SIB2); //@sic R5s230478 sic@
            

            
            // construct the scheduling info list
            v_SchedulingInfoList := cs_SchedulingInfo_NB_4SI_NTN({sibType5_NB_r13});
            v_SIB1.message_.c1.systemInformationBlockType1_r13.nonCriticalExtension := cs_SystemInformationBlockType1_NB_v1350_NTN_r17(p_TrackingAreaCode, cs_SchedulingInfo_NB_c10_c11_c13_r15);
        }
        case (c12) { // new // SIBs 2, 22, 31
            // Add SIBs to v_SI_List
            f_NBIOT_Init_SIBs(31, v_SI_List);
            f_NBIOT_Init_SIBs(22, v_SI_List, p_CellId, p_FrequencyBand, p_OperationMode, p_CellsOnPLMN);
            
            // NTN SIB31 specific config
            v_MIB.message_.partEARFCN_17.earfcn_LSB := fl_NBIOT_Get_EARFCN_2LSB(v_CarrierFreq);
            v_MIB.message_.schedulingInfoSIB1_r13 := 5; //@sic R5-235420 sic@
            v_SIB1.message_.c1.systemInformationBlockType1_r13.cellAccessRelatedInfo_r13.cellBarred_r13 := barred;
            v_SIB2.sib_TypeAndInfo_r13[0].sib2_r13.radioResourceConfigCommon_r13.ntn_ConfigCommon_r17 := cs_508_NTN_ConfigCommon_r17;
            v_SI_List[0] := cs_SystemInformation_Common_NB(v_SIB2); //@sic R5s230478 sic@
            
            
            // construct the scheduling info list
            v_SchedulingInfoList := cs_SchedulingInfo_NB_3SI(-, -, -, {}, -, -, -, {sibType22_NB_r14});
            v_SIB1.message_.c1.systemInformationBlockType1_r13.nonCriticalExtension := cs_SystemInformationBlockType1_NB_v1350_NTN_r17(p_TrackingAreaCode, cs_SchedulingInfo_NB_c12_r15);
        }
        case (c13){ // SIBs 2, 3, 14
            // Add SIBs to v_SI_List
            f_NBIOT_Init_SIBs(31, v_SI_List);
            f_NBIOT_Init_SIBs(3, v_SI_List);
            f_NBIOT_Init_SIBs(14, v_SI_List);

            // NTN SIB31 specific config
            v_MIB.message_.partEARFCN_17.earfcn_LSB := fl_NBIOT_Get_EARFCN_2LSB(v_CarrierFreq);
            v_MIB.message_.schedulingInfoSIB1_r13 := 5; //@sic R5-235420 sic@
            v_SIB1.message_.c1.systemInformationBlockType1_r13.cellAccessRelatedInfo_r13.cellBarred_r13 := barred;
            v_SIB2.sib_TypeAndInfo_r13[0].sib2_r13.radioResourceConfigCommon_r13.ntn_ConfigCommon_r17 := cs_508_NTN_ConfigCommon_r17;
            v_SI_List[0] := cs_SystemInformation_Common_NB(v_SIB2); //@sic R5s230478 sic@
            
            // construct the scheduling info list
            v_SchedulingInfoList := cs_SchedulingInfo_NB_4SI_NTN({sibType14_NB_r13});
            v_SIB1.message_.c1.systemInformationBlockType1_r13.nonCriticalExtension := cs_SystemInformationBlockType1_NB_v1350_NTN_r17(p_TrackingAreaCode, cs_SchedulingInfo_NB_c10_c11_c13_r15);
        }
        case else{
            FatalError(__FILE__, __LINE__, "Combination not supported");
        }
    }

    v_SIB1.message_.c1.systemInformationBlockType1_r13.schedulingInfoList_r13 := v_SchedulingInfoList;
    v_CellBcchInfo := cs_BCCHInfo_NB_C1(v_MIB, v_SIB1, v_SI_List);
    return v_CellBcchInfo;
  }
2.7 Correction to Get and Set Functions
	Function names
	f_NBIOT_CellInfo_GetSIB2 (example where used 22.4.1), f_NBIOT_CellInfo_GetSIB3 (example where used 22.2.4), f_NBIOT_CellInfo_GetSIB5, f_NBIOT_CellInfo_GetSIB14 (example where used 22.4.5),
f_NBIOT_CellInfo_GetSIB22, f_NBIOT_CellInfo_GetSIB23,
f_NBIOT_CellInfo_GetSIB31 (used in f_NBIOT_CellInfo_Get_Koffset) .

f_NBIOT_CellInfo_SetSIB2 (Used in 22.3.1.1), f_NBIOT_CellInfo_SetSIB3,
f_NBIOT_CellInfo_SetSIB4, f_NBIOT_CellInfo_SetSIB5,
f_NBIOT_CellInfo_SetSIB14 (Used in 22.4.8), f_NBIOT_CellInfo_SetSIB22 (Used in 22.4.24),
f_NBIOT_CellInfo_SetSIB23, f_NBIOT_CellInfo_SetSIB31 (Used in fl_NBIOT_Common_Init),
f_NBIOT_CellInfo_ModifySIB31_UL_SyncValidityDuration_r17

	Reason for change
	Functions use harcoded values for getting the index of SIBs from the SI list. 

	Summary of change
	Changed each function to use [2.4] f_NBIOT_CellInfo_Get_SI_Index to get the SI index and then use it to return the correct the SIB.


	TTCN module
	NBIOT_CellInfo.ttcn

	MCC160 Comment
	



Before:
function f_NBIOT_CellInfo_GetSIB2(NBIOT_CellId_Type p_CellId) runs on NBIOT_PTC return SystemInformationBlockType2_NB_r13
  {
    var NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    return v_CellInfo.Sysinfo.BCCH_Info.SIs[0].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib2_r13;
  }

After:
function f_NBIOT_CellInfo_GetSIB2(NBIOT_CellId_Type p_CellId) runs on NBIOT_PTC return SystemInformationBlockType2_NB_r13 // Updated to use function to determine index of where SIB should be WI1233835
  {
    var NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    var integer index := f_NBIOT_CellInfo_Get_SI_Index(p_CellId, 2);
    return v_CellInfo.Sysinfo.BCCH_Info.SIs[index].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib2_r13;
  }

Before:
function f_NBIOT_CellInfo_GetSIB3(NBIOT_CellId_Type p_CellId) runs on NBIOT_PTC return SystemInformationBlockType3_NB_r13
  {
    var NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    return v_CellInfo.Sysinfo.BCCH_Info.SIs[1].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib3_r13;
  }

After:
function f_NBIOT_CellInfo_GetSIB3(NBIOT_CellId_Type p_CellId) runs on NBIOT_PTC return SystemInformationBlockType3_NB_r13 // Updated to use function to determine index of where SIB should be WI1233835
  {
    var NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    var integer index := f_NBIOT_CellInfo_Get_SI_Index(p_CellId, 3);
    return v_CellInfo.Sysinfo.BCCH_Info.SIs[index].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib3_r13;
  }

Before:
function f_NBIOT_CellInfo_GetSIB5(NBIOT_CellId_Type p_CellId) runs on NBIOT_PTC return SystemInformationBlockType5_NB_r13
  {
    var NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    return v_CellInfo.Sysinfo.BCCH_Info.SIs[2].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib5_r13;
  }

After:
  function f_NBIOT_CellInfo_GetSIB5(NBIOT_CellId_Type p_CellId) runs on NBIOT_PTC return SystemInformationBlockType5_NB_r13 // Updated to use function to determine index of where SIB should be WI1233835
  {
    var NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    var integer index := f_NBIOT_CellInfo_Get_SI_Index(p_CellId, 5);
    return v_CellInfo.Sysinfo.BCCH_Info.SIs[index].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib5_r13;
  }

Before:
function f_NBIOT_CellInfo_GetSIB14(NBIOT_CellId_Type p_CellId) runs on NBIOT_PTC return SystemInformationBlockType14_NB_r13
  {
    var NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    return  v_CellInfo.Sysinfo.BCCH_Info.SIs[2].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib14_r13;
  }

After:
  function f_NBIOT_CellInfo_GetSIB14(NBIOT_CellId_Type p_CellId) runs on NBIOT_PTC return SystemInformationBlockType14_NB_r13 // Updated to use function to determine index of where SIB should be WI1233835
  {
    var NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    var integer index := f_NBIOT_CellInfo_Get_SI_Index(p_CellId, 14);
    return  v_CellInfo.Sysinfo.BCCH_Info.SIs[index].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib14_r13;
  }

Before:
function f_NBIOT_CellInfo_GetSIB22(NBIOT_CellId_Type p_CellId) runs on NBIOT_PTC return SystemInformationBlockType22_NB_r14
  {
    var NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    return v_CellInfo.Sysinfo.BCCH_Info.SIs[1].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib22_v1430;
  }

After:
  function f_NBIOT_CellInfo_GetSIB22(NBIOT_CellId_Type p_CellId) runs on NBIOT_PTC return SystemInformationBlockType22_NB_r14 // Updated to use function to determine index of where SIB should be WI1233835
  {
    var NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    var integer index := f_NBIOT_CellInfo_Get_SI_Index(p_CellId, 22);
    return v_CellInfo.Sysinfo.BCCH_Info.SIs[index].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib22_v1430;
  }

Before:
  /*
   * @desc      return the SystemInformationBlockType23_NB_r15 stored in NBIOT_CellInfo
   * @param     p_CellId
   * @return    SystemInformationBlockType22_NB_r15
   * @status
   */
  function f_NBIOT_CellInfo_GetSIB23(NBIOT_CellId_Type p_CellId) runs on NBIOT_PTC return SystemInformationBlockType23_NB_r15
  {
    var NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    return v_CellInfo.Sysinfo.BCCH_Info.SIs[1].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib23_v1530;
  }

After:
/*
   * @desc      return the SystemInformationBlockType23_NB_r15 stored in NBIOT_CellInfo
   * @param     p_CellId
   * @return    SystemInformationBlockType23_NB_r15
   * @status
   */
  function f_NBIOT_CellInfo_GetSIB23(NBIOT_CellId_Type p_CellId) runs on NBIOT_PTC return SystemInformationBlockType23_NB_r15 // Updated to use function to determine index of where SIB should be WI1233835
  {
    var NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    var integer index := f_NBIOT_CellInfo_Get_SI_Index(p_CellId, 23);
    return v_CellInfo.Sysinfo.BCCH_Info.SIs[index].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib23_v1530;
  }

Before:
  function f_NBIOT_CellInfo_GetSIB31(NBIOT_CellId_Type p_CellId) runs on NBIOT_PTC return SystemInformationBlockType31_NB_r17
  {//@sic R5-235420, R5s230550 sic@
    var NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    if (v_CellInfo.Sysinfo.Combination!=c9){
        return v_CellInfo.Sysinfo.BCCH_Info.SIs[1].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib31_v1700;
    }
    else{
        return v_CellInfo.Sysinfo.BCCH_Info.SIs[0].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[1].sib31_v1700;
    }
  }

After:
  function f_NBIOT_CellInfo_GetSIB31(NBIOT_CellId_Type p_CellId) runs on NBIOT_PTC return SystemInformationBlockType31_NB_r17 // Updated to use function to determine index of where SIB should be WI1233835
  {//@sic R5-235420, R5s230550 sic@
    var NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    var integer index := f_NBIOT_CellInfo_Get_SI_Index(p_CellId, 31);

    if (v_CellInfo.Sysinfo.Combination!=c9){
        return v_CellInfo.Sysinfo.BCCH_Info.SIs[index].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib31_v1700;
    }
    else{
        return v_CellInfo.Sysinfo.BCCH_Info.SIs[index].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[1].sib31_v1700;
    }
  }

Before:

function f_NBIOT_CellInfo_SetSIB2(NBIOT_CellId_Type p_CellId,
                                    template (value) SystemInformationBlockType2_NB_r13 p_SIB2) runs on NBIOT_PTC
  {
    var template (value) NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    
    v_CellInfo.Sysinfo.BCCH_Info.SIs[0].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib2_r13 := p_SIB2;
    f_NBIOT_CellInfo_Set(p_CellId, v_CellInfo);
  }

After:
  function f_NBIOT_CellInfo_SetSIB2(NBIOT_CellId_Type p_CellId,
                                    template (value) SystemInformationBlockType2_NB_r13 p_SIB2) runs on NBIOT_PTC // Updated to use function to determine index of where SIB should be WI1233835
  {
    var template (value) NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    var integer index := f_NBIOT_CellInfo_Get_SI_Index(p_CellId, 2);
    
    v_CellInfo.Sysinfo.BCCH_Info.SIs[index].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib2_r13 := p_SIB2;
    f_NBIOT_CellInfo_Set(p_CellId, v_CellInfo);
  }

Before:

  function f_NBIOT_CellInfo_SetSIB3(NBIOT_CellId_Type p_CellId,
                                    template (value) SystemInformationBlockType3_NB_r13 p_SIB3) runs on NBIOT_PTC
  {
    var template (value) NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    
    // SIB3 is only present in combination c2, c3 and c4 (see 36.508 cl. 8.1.4.3.1.2)
    select (valueof(v_CellInfo.Sysinfo.Combination)) {
      case (c2, c3, c4) {
        //The combination initiation can be found in f_NBIOT_InitSystemInformation
        v_CellInfo.Sysinfo.BCCH_Info.SIs[1].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib3_r13 := p_SIB3;
        f_NBIOT_CellInfo_Set(p_CellId, v_CellInfo);
      }
      case else {
        FatalError(__FILE__, __LINE__, "Invalid System Combination");
      }
    }
  }

After:
  function f_NBIOT_CellInfo_SetSIB3(NBIOT_CellId_Type p_CellId,
                                    template (value) SystemInformationBlockType3_NB_r13 p_SIB3) runs on NBIOT_PTC // Updated to use function to determine index of where SIB should be WI1233835
  {
    var template (value) NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    var integer index := f_NBIOT_CellInfo_Get_SI_Index(p_CellId, 3);
    
    // SIB3 is only present in combination c2, c3 and c4 (see 36.508 cl. 8.1.4.3.1.2)
    select (valueof(v_CellInfo.Sysinfo.Combination)) {
      case (c2, c3, c4, c9, c10, c11, c13) {
        //The combination initiation can be found in f_NBIOT_InitSystemInformation
        v_CellInfo.Sysinfo.BCCH_Info.SIs[index].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib3_r13 := p_SIB3;
        f_NBIOT_CellInfo_Set(p_CellId, v_CellInfo);
      }
      case else {
        FatalError(__FILE__, __LINE__, "Invalid System Combination");
      }
    }
  }

Before:

function f_NBIOT_CellInfo_SetSIB4(NBIOT_CellId_Type p_CellId,
                                    template (value) SystemInformationBlockType4_NB_r13 p_Sib4) runs on NBIOT_PTC
  {
    var template (value) NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    
    // SIB4 is only present in combination c2 (see 36.508 cl. 8.1.4.3.1.2)
    select (valueof(v_CellInfo.Sysinfo.Combination)) {
      case (c2) {
        //The combination initiation can be found in f_NBIOT_InitSystemInformation
        v_CellInfo.Sysinfo.BCCH_Info.SIs[2].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib4_r13 := p_Sib4;
        f_NBIOT_CellInfo_Set(p_CellId, v_CellInfo);
      }
      case else {
        FatalError(__FILE__, __LINE__, "Invalid System Combination");
      }
    }
  }

After:
  function f_NBIOT_CellInfo_SetSIB4(NBIOT_CellId_Type p_CellId,
                                    template (value) SystemInformationBlockType4_NB_r13 p_Sib4) runs on NBIOT_PTC // Updated to use function to determine index of where SIB should be WI1233835
  {
    var template (value) NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    var integer index := f_NBIOT_CellInfo_Get_SI_Index(p_CellId, 4);
    
    // SIB4 is only present in combination c2 (see 36.508 cl. 8.1.4.3.1.2)
    select (valueof(v_CellInfo.Sysinfo.Combination)) {
      case (c2, c10) {
        //The combination initiation can be found in f_NBIOT_InitSystemInformation
        v_CellInfo.Sysinfo.BCCH_Info.SIs[index].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib4_r13 := p_Sib4;
        f_NBIOT_CellInfo_Set(p_CellId, v_CellInfo);
      }
      case else {
        FatalError(__FILE__, __LINE__, "Invalid System Combination");
      }
    }
  }

Before:

function f_NBIOT_CellInfo_SetSIB5(NBIOT_CellId_Type p_CellId,
                                    template (value) SystemInformationBlockType5_NB_r13 p_SIB5) runs on NBIOT_PTC
  {
    var template (value) NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    
    // SIB5 is only present in combination c3 (see 36.508 cl. 8.1.4.3.1.2)
    select (valueof(v_CellInfo.Sysinfo.Combination)) {
      case (c3) {
        v_CellInfo.Sysinfo.BCCH_Info.SIs[2].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib5_r13 := p_SIB5;
        f_NBIOT_CellInfo_Set(p_CellId, v_CellInfo);
      }
      case else {
        FatalError(__FILE__, __LINE__, "Invalid System Combination");
      }
    }
  }

After:
  function f_NBIOT_CellInfo_SetSIB5(NBIOT_CellId_Type p_CellId,
                                    template (value) SystemInformationBlockType5_NB_r13 p_SIB5) runs on NBIOT_PTC // Updated to use function to determine index of where SIB should be WI1233835
  {
    var template (value) NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    var integer index := f_NBIOT_CellInfo_Get_SI_Index(p_CellId, 5);
    
    // SIB5 is only present in combination c3 (see 36.508 cl. 8.1.4.3.1.2)
    select (valueof(v_CellInfo.Sysinfo.Combination)) {
      case (c3, c11) {
        v_CellInfo.Sysinfo.BCCH_Info.SIs[index].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib5_r13 := p_SIB5;
        f_NBIOT_CellInfo_Set(p_CellId, v_CellInfo);
      }
      case else {
        FatalError(__FILE__, __LINE__, "Invalid System Combination");
      }
    }
  }

Before:

function f_NBIOT_CellInfo_SetSIB14(NBIOT_CellId_Type p_CellId,
                                     template (value) SystemInformationBlockType14_NB_r13 p_SIB14) runs on NBIOT_PTC
  {
    var template (value) NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    
    // SIB14 is only present in combination c4 (see 36.508 cl. 8.1.4.3 Table 8.1.4.3.1.1-1 for more information)
    select (valueof(v_CellInfo.Sysinfo.Combination)) {
      case (c4) {
        v_CellInfo.Sysinfo.BCCH_Info.SIs[2].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib14_r13 := p_SIB14;
        f_NBIOT_CellInfo_Set(p_CellId, v_CellInfo);
      }
      case else {
        FatalError (__FILE__, __LINE__, "Invalid System Combination");
      }
    }
  }

After:
    function f_NBIOT_CellInfo_SetSIB14(NBIOT_CellId_Type p_CellId,
                                     template (value) SystemInformationBlockType14_NB_r13 p_SIB14) runs on NBIOT_PTC // Updated to use function to determine index of where SIB should be WI1233835
  {
    var template (value) NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    var integer index := f_NBIOT_CellInfo_Get_SI_Index(p_CellId, 14);
    
    // SIB14 is only present in combination c4 (see 36.508 cl. 8.1.4.3 Table 8.1.4.3.1.1-1 for more information)
    select (valueof(v_CellInfo.Sysinfo.Combination)) {
      case (c4, c13) {
        v_CellInfo.Sysinfo.BCCH_Info.SIs[index].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib14_r13 := p_SIB14;
        f_NBIOT_CellInfo_Set(p_CellId, v_CellInfo);
      }
      case else {
        FatalError (__FILE__, __LINE__, "Invalid System Combination");
      }
    }
  }

Before:

    function f_NBIOT_CellInfo_SetSIB22(NBIOT_CellId_Type p_CellId,
                                    template (value) SystemInformationBlockType22_NB_r14 p_SIB22) runs on NBIOT_PTC
  {
    var template (value) NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    
    // SIB22 is only present in combination c6 (see 36.508 cl. 8.1.4.3.1.2)
    select (valueof(v_CellInfo.Sysinfo.Combination)) {
      case (c6) {
        //The combination initiation can be found in f_NBIOT_InitSystemInformation
        v_CellInfo.Sysinfo.BCCH_Info.SIs[1].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib22_v1430 := p_SIB22;
        f_NBIOT_CellInfo_Set(p_CellId, v_CellInfo);
      }
      case else {
        FatalError(__FILE__, __LINE__, "Invalid System Combination");
      }
    }
  }

After:
    function f_NBIOT_CellInfo_SetSIB22(NBIOT_CellId_Type p_CellId,
                                    template (value) SystemInformationBlockType22_NB_r14 p_SIB22) runs on NBIOT_PTC // Updated to use function to determine index of where SIB should be WI1233835
  {
    var template (value) NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    var integer index := f_NBIOT_CellInfo_Get_SI_Index(p_CellId, 22);
      
    // SIB22 is only present in combination c6 (see 36.508 cl. 8.1.4.3.1.2)
    select (valueof(v_CellInfo.Sysinfo.Combination)) {
      case (c6, c7, c12) {
        //The combination initiation can be found in f_NBIOT_InitSystemInformation
        v_CellInfo.Sysinfo.BCCH_Info.SIs[index].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib22_v1430 := p_SIB22;
        f_NBIOT_CellInfo_Set(p_CellId, v_CellInfo);
      }
      case else {
        FatalError(__FILE__, __LINE__, "Invalid System Combination");
      }
    }
  }

Before:

  function f_NBIOT_CellInfo_SetSIB23(NBIOT_CellId_Type p_CellId,
                                    template (value) SystemInformationBlockType23_NB_r15 p_SIB23) runs on NBIOT_PTC
  {
    var template (value) NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    
    // SIB23 is only present in combination c7 (see 36.508 cl. 8.1.4.3.1.2)
    select (valueof(v_CellInfo.Sysinfo.Combination)) {
      case (c7) {
        //The combination initiation can be found in f_NBIOT_InitSystemInformation
        v_CellInfo.Sysinfo.BCCH_Info.SIs[1].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib23_v1530 := p_SIB23;
        f_NBIOT_CellInfo_Set(p_CellId, v_CellInfo);
      }
      case else {
        FatalError(__FILE__, __LINE__, "Invalid System Combination");
      }
    }
  }

After:
    function f_NBIOT_CellInfo_SetSIB23(NBIOT_CellId_Type p_CellId,
                                    template (value) SystemInformationBlockType23_NB_r15 p_SIB23) runs on NBIOT_PTC // Updated to use function to determine index of where SIB should be WI1233835
  {
    var template (value) NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    var integer index := f_NBIOT_CellInfo_Get_SI_Index(p_CellId, 23);
    
    // SIB23 is only present in combination c7 (see 36.508 cl. 8.1.4.3.1.2)
    select (valueof(v_CellInfo.Sysinfo.Combination)) {
      case (c7) {
        //The combination initiation can be found in f_NBIOT_InitSystemInformation
        v_CellInfo.Sysinfo.BCCH_Info.SIs[index].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib23_v1530 := p_SIB23;
        f_NBIOT_CellInfo_Set(p_CellId, v_CellInfo);
      }
      case else {
        FatalError(__FILE__, __LINE__, "Invalid System Combination");
      }
    }
  }


Before:

  function f_NBIOT_CellInfo_SetSIB31(NBIOT_CellId_Type p_CellId,
                                    template (value) SystemInformationBlockType31_NB_r17 p_SIB31) runs on NBIOT_PTC
  {//@sic R5-235420, R5s230550 sic@
    var template (value) NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
      
    if (valueof(v_CellInfo.Sysinfo.Combination) != c9){
        v_CellInfo.Sysinfo.BCCH_Info.SIs[0].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib31_v1700 := p_SIB31;
    }
    else{
        v_CellInfo.Sysinfo.BCCH_Info.SIs[0].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[1].sib31_v1700 := p_SIB31;
    }
    f_NBIOT_CellInfo_Set(p_CellId, v_CellInfo);
  }

After:
    function f_NBIOT_CellInfo_SetSIB31(NBIOT_CellId_Type p_CellId,
                                    template (value) SystemInformationBlockType31_NB_r17 p_SIB31) runs on NBIOT_PTC // Updated to use function to determine index of where SIB should be WI1233835
  {//@sic R5-235420, R5s230550 sic@
    var template (value) NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    var integer index := f_NBIOT_CellInfo_Get_SI_Index(p_CellId, 31);
        
    if (valueof(v_CellInfo.Sysinfo.Combination) != c9){
        v_CellInfo.Sysinfo.BCCH_Info.SIs[index].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib31_v1700 := p_SIB31;
    }
    else{
        v_CellInfo.Sysinfo.BCCH_Info.SIs[index].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[1].sib31_v1700 := p_SIB31;
    }
    f_NBIOT_CellInfo_Set(p_CellId, v_CellInfo);
  }

Before:

    function f_NBIOT_CellInfo_ModifySIB31_UL_SyncValidityDuration_r17(NBIOT_CellId_Type p_CellId,
                                    template (value) SystemInformationBlockType31_NB_r17.servingSatelliteInfo_r17.ul_SyncValidityDuration_r17 p_UL_SyncValidityDuration_r17) runs on NBIOT_PTC
  {
    var template (value) NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    
    // SIB31-NB is only present in combinations c8..c13 (see 36.508 cl. 8.1.4.3.1.2)
    select (valueof(v_CellInfo.Sysinfo.Combination)) {
      case (c8, c10, c11, c12, c13) {
        //The combination initiation can be found in f_NBIOT_InitSystemInformation
        v_CellInfo.Sysinfo.BCCH_Info.SIs[1].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib31_v1700.servingSatelliteInfo_r17.ul_SyncValidityDuration_r17 := p_UL_SyncValidityDuration_r17;
      }
      case (c9) {
        v_CellInfo.Sysinfo.BCCH_Info.SIs[2].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib31_v1700.servingSatelliteInfo_r17.ul_SyncValidityDuration_r17 := p_UL_SyncValidityDuration_r17;
      }
      case else {
        FatalError(__FILE__, __LINE__, "Invalid System Combination");
      }
    }
    f_NBIOT_CellInfo_Set(p_CellId, v_CellInfo);
  }

After:
    function f_NBIOT_CellInfo_ModifySIB31_UL_SyncValidityDuration_r17(NBIOT_CellId_Type p_CellId,
                                    template (value) SystemInformationBlockType31_NB_r17.servingSatelliteInfo_r17.ul_SyncValidityDuration_r17 p_UL_SyncValidityDuration_r17) runs on NBIOT_PTC // Updated to use function to determine index of where SIB should be WI1233835
  {
    var template (value) NBIOT_CellInfo_Type v_CellInfo := f_NBIOT_CellInfo_Get(p_CellId);
    var integer index := f_NBIOT_CellInfo_Get_SI_Index(p_CellId, 31);
    
    // SIB31-NB is only present in combinations c8..c13 (see 36.508 cl. 8.1.4.3.1.2)
    select (valueof(v_CellInfo.Sysinfo.Combination)) {
      case (c8, c10, c11, c12, c13) {
        //The combination initiation can be found in f_NBIOT_InitSystemInformation
        v_CellInfo.Sysinfo.BCCH_Info.SIs[index].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[0].sib31_v1700.servingSatelliteInfo_r17.ul_SyncValidityDuration_r17 := p_UL_SyncValidityDuration_r17;
      }
      case (c9) {
        v_CellInfo.Sysinfo.BCCH_Info.SIs[index].message_.c1.systemInformation_r13.criticalExtensions.systemInformation_r13.sib_TypeAndInfo_r13[1].sib31_v1700.servingSatelliteInfo_r17.ul_SyncValidityDuration_r17 := p_UL_SyncValidityDuration_r17;
      }
      case else {
        FatalError(__FILE__, __LINE__, "Invalid System Combination");
      }
    }
    f_NBIOT_CellInfo_Set(p_CellId, v_CellInfo);
  }


